

NAG Fortran Library Routine Document

F08QYF (CTRSNA/ZTRSNA)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08QYF (CTRSNA/ZTRSNA) estimates condition numbers for specified eigenvalues and/or right eigenvectors of a complex upper triangular matrix.

2 Specification

```

SUBROUTINE F08QYF(JOB, HOWMNY, SELECT, N, T, LDT, VL, LDVL, VR, LDVR, S,
1              SEP, MM, M, WORK, LDWORK, RWORK, INFO)
ENTRY          ctrsna (JOB, HOWMNY, SELECT, N, T, LDT, VL, LDVL, VR, LDVR, S,
1              SEP, MM, M, WORK, LDWORK, RWORK, INFO)
INTEGER        N, LDT, LDVL, LDVR, MM, M, LDWORK, INFO
real          S(*), SEP(*), RWORK(*)
complex      T(LDT,*), VL(LDVL,*), VR(LDVR,*), WORK(LDWORK,*)
LOGICAL        SELECT(*)
CHARACTER*1    JOB, HOWMNY

```

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

This routine estimates condition numbers for specified eigenvalues and/or right eigenvectors of a complex upper triangular matrix T . These are the same as the condition numbers of the eigenvalues and right eigenvectors of an original matrix $A = ZTZ^H$ (with unitary Z), from which T may have been derived.

F08QYF computes the reciprocal of the condition number of an eigenvalue λ_i as

$$s_i = \frac{|v^H u|}{\|u\|_E \|v\|_E},$$

where u and v are the right and left eigenvectors of T , respectively, corresponding to λ_i . This reciprocal condition number always lies between zero (i.e., ill-conditioned) and one (i.e., well-conditioned).

An approximate error estimate for a computed eigenvalue λ_i is then given by

$$\frac{\epsilon \|T\|}{s_i},$$

where ϵ is the ***machine precision***.

To estimate the reciprocal of the condition number of the right eigenvector corresponding to λ_i , the routine first calls F08QTF (CTREXC/ZTREXC) to reorder the eigenvalues so that λ_i is in the leading position:

$$T = Q \begin{pmatrix} \lambda_i & c^H \\ 0 & T_{22} \end{pmatrix} Q^H.$$

The reciprocal condition number of the eigenvector is then estimated as sep_i , the smallest singular value of the matrix $(T_{22} - \lambda_i I)$. This number ranges from zero (i.e., ill-conditioned) to very large (i.e., well-conditioned).

An approximate error estimate for a computed right eigenvector u corresponding to λ_i is then given by

$$\frac{\epsilon \|T\|}{sep_i}.$$

4 References

Golub G H and van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

- 1: JOB – CHARACTER*1 *Input*
On entry: indicates whether condition numbers are required for eigenvalues and/or eigenvectors, as follows:
 if JOB = 'E', then condition numbers for eigenvalues only are computed;
 if JOB = 'V', then condition numbers for eigenvectors only are computed;
 if JOB = 'B', then condition numbers for both eigenvalues and eigenvectors are computed.
Constraint: JOB = 'E', 'V' or 'B'.
- 2: HOWMNY – CHARACTER*1 *Input*
On entry: indicates how many condition numbers are to be computed, as follows:
 if HOWMNY = 'A', then condition numbers for all eigenpairs are computed;
 if HOWMNY = 'S', then condition numbers for selected eigenpairs (as specified by SELECT) are computed.
Constraint: HOWMNY = 'A' or 'S'.
- 3: SELECT(*) – LOGICAL array *Input*
Note: the dimension of the array SELECT must be at least $\max(1, N)$ if HOWMNY = 'S' and at least 1 otherwise.
On entry: SELECT specifies the eigenpairs for which condition numbers are to be computed if HOWMNY = 'S'. To select condition numbers for the eigenpair corresponding to the eigenvalue λ_j , SELECT(*j*) must be set to .TRUE..
 SELECT is not referenced if HOWMNY = 'A'.
- 4: N – INTEGER *Input*
On entry: *n*, the order of the matrix *T*.
Constraint: $N \geq 0$.
- 5: T(LDT,*) – **complex** array *Input*
Note: the second dimension of the array T must be at least $\max(1, N)$.
On entry: the *n* by *n* upper triangular matrix *T*, as returned by F08PSF (CHSEQR/ZHSEQR).
- 6: LDT – INTEGER *Input*
On entry: the first dimension of the array T as declared in the (sub)program from which F08QYF (CTRSNA/ZTRSNA) is called.
Constraint: $LDT \geq \max(1, N)$.
- 7: VL(LDVL,*) – **complex** array *Input*
Note: the second dimension of the array VL must be at least $\max(1, N)$ if JOB = 'E' or 'B' and at least 1 if JOB = 'V'.
On entry: if JOB = 'E' or 'B', VL must contain the left eigenvectors of *T* (or of any matrix QTQ^H with *Q* unitary) corresponding to the eigenpairs specified by HOWMNY and SELECT. The

eigenvectors **must** be stored in consecutive columns of VL, as returned by F08QXF (CTREVC/ZTREVC) or F08PXF (CHSEIN/ZHSEIN).

VL is not referenced if JOB = 'V'.

- 8: LDVL – INTEGER *Input*

On entry: the first dimension of the array VL as declared in the (sub)program from which F08QYF (CTRSNA/ZTRSNA) is called.

Constraints:

$$\begin{aligned} \text{LDVL} &\geq \max(1, N) \text{ if JOB = 'E' or 'B',} \\ \text{LDVL} &\geq 1 \text{ if JOB = 'V'.} \end{aligned}$$

- 9: VR(LDVR,*) – **complex** array *Input*

Note: the second dimension of the array VR must be at least $\max(1, \text{MM})$ if JOB = 'E' or 'B' and at least 1 if JOB = 'V'.

On entry: if JOB = 'E' or 'B', VR must contain the right eigenvectors of T (or of any matrix QTQ^H with Q unitary) corresponding to the eigenpairs specified by HOWMNY and SELECT. The eigenvectors **must** be stored in consecutive columns of VR, as returned by F08QXF (CTREVC/ZTREVC) or F08PXF (CHSEIN/ZHSEIN).

VR is not referenced if JOB = 'V'.

- 10: LDVR – INTEGER *Input*

On entry: the first dimension of the array VR as declared in the (sub)program from which F08QYF (CTRSNA/ZTRSNA) is called.

Constraints:

$$\begin{aligned} \text{LDVR} &\geq \max(1, N) \text{ if JOB = 'E' or 'B',} \\ \text{LDVR} &\geq 1 \text{ if JOB = 'V'.} \end{aligned}$$

- 11: S(*) – **real** array *Output*

Note: the dimension of the array S must be at least $\max(1, \text{MM})$ if JOB = 'E' or 'B' and at least 1 if JOB = 'V'.

On exit: the reciprocal condition numbers of the selected eigenvalues if JOB = 'E' or 'B', stored in consecutive elements of the array. Thus $S(j)$, $\text{SEP}(j)$ and the j th columns of VL and VR all correspond to the same eigenpair (but not in general the j th eigenpair unless all eigenpairs have been selected).

S is not referenced if JOB = 'V'.

- 12: SEP(*) – **real** array *Output*

Note: the dimension of the array SEP must be at least $\max(1, \text{MM})$ if JOB = 'V' or 'B' and at least 1 if JOB = 'E'.

On exit: the estimated reciprocal condition numbers of the selected right eigenvectors if JOB = 'V' or 'B', stored in consecutive elements of the array.

SEP is not referenced if JOB = 'E'.

- 13: MM – INTEGER *Input*

On entry: the number of elements in the arrays S and SEP, and the number of columns in the arrays VL and VR (if used). The precise number required, m , is n if HOWMNY = 'A'; if HOWMNY = 'S', m is the number of selected eigenpairs (see SELECT), in which case $0 \leq m \leq n$.

Constraint: $\text{MM} \geq m$.

- 14: M – INTEGER *Output*
On exit: m , the number of selected eigenpairs. If HOWMNY='A', M is set to n .
- 15: WORK(LDWORK,*) – **complex** array *Workspace*
Note: the second dimension of the array WORK must be at least $\max(1, N + 1)$ if JOB = 'V' or 'B' and at least 1 if JOB = 'E'.
 WORK is not referenced if JOB = 'E'.
- 16: LDWORK – INTEGER *Input*
On entry: the first dimension of the array WORK as declared in the (sub)program from which F08QYF (CTRSNA/ZTRSNA) is called.
Constraints:
 $LDWORK \geq \max(1, N)$ if JOB = 'V' or 'B',
 $LDWORK \geq 1$ if JOB = 'E'.
- 17: RWORK(*) – **real** array *Workspace*
Note: the dimension of the array RWORK must be at least $\max(1, N)$.
- 18: INFO – INTEGER *Output*
On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If $INFO = -i$, the i th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

The computed values sep_i may overestimate the true value, but seldom by a factor of more than 3.

8 Further Comments

The real analogue of this routine is F08QLF (STRSNA/DTRSNA).

9 Example

To compute approximate error estimates for all the eigenvalues and right eigenvectors of the matrix T , where

$$T = \begin{pmatrix} -6.0004 - 6.9999i & 0.3637 - 0.3656i & -0.1880 + 0.4787i & 0.8785 - 0.2539i \\ 0.0000 + 0.0000i & -5.0000 + 2.0060i & -0.0307 - 0.7217i & -0.2290 + 0.1313i \\ 0.0000 + 0.0000i & 0.0000 + 0.0000i & 7.9982 - 0.9964i & 0.9357 + 0.5359i \\ 0.0000 + 0.0000i & 0.0000 + 0.0000i & 0.0000 + 0.0000i & 3.0023 - 3.9998i \end{pmatrix}.$$

9.1 Program Text

Note: the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      F08QYF Example Program Text
*      Mark 16 Release. NAG Copyright 1992.
*      .. Parameters ..
INTEGER          NIN, NOUT
PARAMETER        (NIN=5,NOUT=6)
INTEGER          NMAX, LDT, LDWORK, LDVL, LDVR
PARAMETER        (NMAX=8,LDT=NMAX,LDWORK=NMAX,LDVL=NMAX,LDVR=NMAX)
*      .. Local Scalars ..
real            EPS, TNORM
INTEGER          I, INFO, J, M, N
*      .. Local Arrays ..
complex         T(LDT,NMAX), VL(LDVL,NMAX), VR(LDVR,NMAX),
+                WORK(LDWORK,NMAX+6)
real            RWORK(NMAX), S(NMAX), SEP(NMAX)
LOGICAL          SELECT(1)
*      .. External Functions ..
real            F06UAF, X02AJF
EXTERNAL         F06UAF, X02AJF
*      .. External Subroutines ..
EXTERNAL         ctrevc, ctrsna
*      .. Executable Statements ..
WRITE (NOUT,*) 'F08QYF Example Program Results'
*      Skip heading in data file
READ (NIN,*)
READ (NIN,*) N
IF (N.LE.NMAX) THEN

*
*      Read T from data file
*
READ (NIN,*) ((T(I,J),J=1,N),I=1,N)

*
*      Calculate the left and right eigenvectors of T
*
CALL ctrevc('Both','All',SELECT,N,T,LDT,VL,LDVL,VR,LDVR,N,M,
+          WORK,RWORK,INFO)

*
*      Estimate condition numbers for all the eigenvalues and right
*      eigenvectors of T
*
CALL ctrsna('Both','All',SELECT,N,T,LDT,VL,LDVL,VR,LDVR,S,SEP,
+          N,M,WORK,LDWORK,RWORK,INFO)

*
*      Print condition numbers of eigenvalues and right eigenvectors
*
WRITE (NOUT,*)
WRITE (NOUT,*) 'S'
WRITE (NOUT,99999) (S(I),I=1,M)
WRITE (NOUT,*)
WRITE (NOUT,*) 'SEP'
WRITE (NOUT,99999) (SEP(I),I=1,M)

*
*      Calculate approximate error estimates (using the 1-norm)
*
EPS = X02AJF()
TNORM = F06UAF('1-norm',N,N,T,LDT,RWORK)
WRITE (NOUT,*)
WRITE (NOUT,*) 'Approximate error estimates for eigenvalues ',
+ 'of T (machine-dependent)'
WRITE (NOUT,99999) (EPS*TNORM/S(I),I=1,M)
WRITE (NOUT,*)
WRITE (NOUT,*) 'Approximate error estimates for right ',
+ 'eigenvectors of T (machine-dependent)'
WRITE (NOUT,99999) (EPS*TNORM/SEP(I),I=1,M)
END IF
STOP
```

```

*
99999 FORMAT ((3X,1P,7E11.1))
      END

```

9.2 Program Data

F08QYF Example Program Data

```

4                                     :Value of N
(-6.0004,-6.9999) ( 0.3637,-0.3656) (-0.1880, 0.4787) ( 0.8785,-0.2539)
( 0.0000, 0.0000) (-5.0000, 2.0060) (-0.0307,-0.7217) (-0.2290, 0.1313)
( 0.0000, 0.0000) ( 0.0000, 0.0000) ( 7.9982,-0.9964) ( 0.9357, 0.5359)
( 0.0000, 0.0000) ( 0.0000, 0.0000) ( 0.0000, 0.0000) ( 3.0023,-3.9998)
                                     :End of matrix T

```

9.3 Program Results

F08QYF Example Program Results

```

S
      9.9E-01      1.0E-00      9.8E-01      9.8E-01

```

```

SEP
      8.4E+00      8.0E+00      5.8E+00      5.8E+00

```

```

Approximate error estimates for eigenvalues of T (machine-dependent)
      1.0E-15      1.0E-15      1.1E-15      1.1E-15

```

```

Approximate error estimates for right eigenvectors of T (machine-dependent)
      1.2E-16      1.3E-16      1.8E-16      1.8E-16

```
